Gibbs-Bagdasaryan-Baez reduction

Version 2: 10 February 2015: Added checks that points *W*, *X*, *Y* are inside the triangle *B*′

This notebook analyses a “universal cover” devised by Gibbs, Bagdasaryan and Baez for Lebesgue’s Universal Covering Problem.

See http://arxiv.org/abs/1502.01251 for details.

Original, unrotated hexagon of width 1, circumscribed radius $\frac{1}{\sqrt{3}}$. Vertices are listed in clockwise order starting from $\left(0, \frac{1}{\sqrt{3}}\right)$.

In[1]:=
```
hrad = 1/Sqrt[3];
```

In[2]:=
```
H1 = Table[hrad {Sin[π i/3], Cos[π i/3]}, {i, 0, 5}]
```

Out[2]= $\left\{\left\{0, \frac{1}{\sqrt{3}}\right\}, \left\{\frac{1}{2}, \frac{1}{2\sqrt{3}}\right\}, \left\{\frac{1}{2}, -\frac{1}{2\sqrt{3}}\right\}, \left\{0, -\frac{1}{\sqrt{3}}\right\}, \left\{-\frac{1}{2}, -\frac{1}{2\sqrt{3}}\right\}, \left\{-\frac{1}{2}, \frac{1}{2\sqrt{3}}\right\}\right\}$

Rotated hexagon: the original hexagon, rotated *counterclockwise*, by $\frac{\pi}{6}+\sigma$. Here we use $s = \sin(\sigma)$ and $c = \cos(\sigma)$ rather than explicit trigonometric functions.

In[3]:=
```
rotH[s_, c_] = Map[{{c, -s}, {s, c}}.# &, TrigExpand[
    Table[hrad {Sin[π i/3 - π/6], Cos[π i/3 - π/6]}, {i, 0, 5}]]]
```

Out[3]= $\left\{\left\{-\frac{c}{2\sqrt{3}}-\frac{s}{2}, \frac{c}{2}-\frac{s}{2\sqrt{3}}\right\}, \left\{\frac{c}{2\sqrt{3}}-\frac{s}{2}, \frac{c}{2}+\frac{s}{2\sqrt{3}}\right\}, \left\{\frac{c}{\sqrt{3}}, \frac{s}{\sqrt{3}}\right\}, \left\{\frac{c}{2\sqrt{3}}+\frac{s}{2}, -\frac{c}{2}+\frac{s}{2\sqrt{3}}\right\}, \left\{-\frac{c}{2\sqrt{3}}+\frac{s}{2}, -\frac{c}{2}-\frac{s}{2\sqrt{3}}\right\}, \left\{-\frac{c}{\sqrt{3}}, -\frac{s}{\sqrt{3}}\right\}\right\}$

In[4]:=
```
csAssume = 0 ≤ s < Sin[π/6] && Cos[π/6] < c ≤ 1
```

Out[4]= $0 \le s < \frac{1}{2}\ \&\&\ \frac{\sqrt{3}}{2} < c \le 1$

Routines to tidy up expressions containing cosines and sines

In[5]:=
```
cst0[x_] :=
  (MapAll[ExpandAll, x] //. {s^n_ /; Mod[n, 2] == 0 :> (1 - c^2)^(n/2),
     s^n_ /; Mod[n, 2] == 1 :> s (1 - c^2)^((n-1)/2)})
```

In[6]:=
```
Map[cst0, Table[s^i, {i, 1, 10}]]
```

Out[6]= $\{s,\ 1-c^2,\ (1-c^2)\,s,\ (1-c^2)^2,\ (1-c^2)^2\,s,\ (1-c^2)^3,\ (1-c^2)^3\,s,\ (1-c^2)^4,\ (1-c^2)^4\,s,\ (1-c^2)^5\}$

In[7]:=
```
cst[x_, n_] := Nest[cst0, x, n]
```

In[8]:=
```
cstNumerator[x_, n_] := Module[{t}, t = Together[x];
   Factor[cst[Numerator[t], n], Extension → Automatic] /
    Simplify[Denominator[t]]]
```

Point of intersection of two line segments, with endpoints *a*, *b* for first line and *c*, *d* for second.

In[9]:=
```
ils[{ax_, ay_}, {bx_, by_}, {cx_, cy_}, {dx_, dy_}] =
 Module[{p1, p2},
  p1 = λ {ax, ay} + (1 - λ) {bx, by};
  p2 = μ {cx, cy} + (1 - μ) {dx, dy};
  Simplify[p1 /. Solve[p1 == p2, {λ, μ}][[1]]]]
```

Out[9]=
```
{(-bx cy dx + ay bx (-cx + dx) +
     bx cx dy + ax (by cx - by dx + cy dx - cx dy)) /
   (by (cx - dx) + ay (-cx + dx) + (ax - bx) (cy - dy)),
  (by (-cy dx + ax (cy - dy) + cx dy) +
     ay (cy dx - cx dy + bx (-cy + dy))) /
   (by (cx - dx) + ay (-cx + dx) + (ax - bx) (cy - dy))}
```

Points of intersection between the two hexagons. Each side of the original hexagon intersects both the same side of the counterclockwise-rotated hexagon, and the “next” side in clockwise order.

In[10]:=
```
pih[s_, c_] = Module[{ip, j, jp, rh},
   rh = rotH[s, c];
   Flatten[Table[
     ip = Mod[i + 1, 6, 1];
     j = Mod[i + k, 6, 1];
     jp = Mod[j + 1, 6, 1];
     Simplify[cst[Simplify[
        ils[H1[[i]], H1[[ip]], rh[[j]], rh[[jp]]],
        csAssume], 2]], {i, 1, 6}, {k, 0, 1}], 1]];
```

The vertices of the unrotated hexagon are named A_1, B_1, ..., F_1, while the first intersection point clockwise from each vertex is A_2, B_2, ..., F_2 and the first intersection point counterclockwise from each vertex is A_3, B_3, ... F_3.

Some of these are specially named in the paper: point $N = E_3$ and point $O = C_2$ are the 8th and 5th of the intersection points.

In[11]:=
```
Npt[s_, c_] = pih[s, c][[8]]
```

Out[11]= $\left\{\frac{-3+\sqrt{3}\,c+3\,s}{2\,\sqrt{3}\,c-6\,s},\ \frac{-3\,c+\sqrt{3}\,(1+s)}{2\,\sqrt{3}\,c-6\,s}\right\}$

In[12]:=
```
Opt[s_, c_] = pih[s, c][[5]]
```

Out[12]= $\left\{\frac{-c+\sqrt{3}\,(1+s)}{2\left(c+\sqrt{3}\,s\right)},\ -\frac{3\,c+\sqrt{3}\,(-1+s)}{2\,\sqrt{3}\,c+6\,s}\right\}$

In[13]:=
```
letters = {"A", "B", "C", "D", "E", "F"};
```

In[14]:=
```
H1Labels = Map[Subscript[#, 1] &, letters]
```

Out[14]= $\{A_1, B_1, C_1, D_1, E_1, F_1\}$

In[15]:=
```
pihLabels =
  Table[Subscript[letters[[Mod[Floor[(i + 2) / 2], 6, 1]]],
     Mod[i, 2, 1] + 1], {i, 1, 12}] /.
   {"E"_3 → "N=E"_3, "C"_2 → "O=C"_2}
```

Out[15]= $\{A_2, B_3, B_2, C_3, O{=}C_2, D_3, D_2, N{=}E_3, E_2, F_3, F_2, A_3\}$

In[16]:=
```
drawHexagons[σ_] := Module[{s, c, pih0},
   s = Sin[σ];
   c = Cos[σ];
   pih0 = pih[s, c];
   Graphics[
    {Table[
      {Line[{H1[[i]], H1[[Mod[i + 1, 6, 1]]]}],
       Line[{pih0[[2 * i]], pih0[[Mod[2 * i + 1, 12, 1]]]}],
       Text[H1Labels[[i]], 1.1 H1[[i]]]
      }
      , {i, 1, 6}],
     Table[
      Text[pihLabels[[i]], 1.1 pih0[[i]]], {i, 1, 12}]
    }, BaseStyle → 18, ImageSize → {500, 500}
   ]]
```

In[17]:=
```
drawHexagons[1.3 Degree]
```

Out[17]=



Area of an arbitrary polygon

In[18]:=
```
areaPoly[p_] := Module[{n, p1, p2},
  n = Length[p];
  Sum[
   p1 = p[[i]];
   p2 = p[[Mod[i + 1, n, 1]]];
   (p1[[2]] p2[[1]] - p1[[1]] p2[[2]]) / 2,
   {i, 1, n}]
 ]
```

Area between a chord of length *c* and a circle of radius 1

In[19]:= chordArea[c_] = $\text{ArcSin}\left[\frac{c}{2}\right] - \frac{c}{2}\sqrt{1-\left(\frac{c}{2}\right)^2}$

Out[19]= $-\frac{1}{2}\,c\,\sqrt{1-\frac{c^2}{4}}+\text{ArcSin}\left[\frac{c}{2}\right]$

Find the points on the original hexagon where circles of radius 1 centred at *O* and *N* are tangent to the hexagon.

In[20]:=
```
Otan[s_, c_] = Simplify[H1[[1]] + Opt[s, c] - H1[[3]]]
```

Out[20]=
$$\left\{\frac{\sqrt{3}-2\,c}{2\,c+2\,\sqrt{3}\,s},\ \frac{\sqrt{3}\,(1+2\,s)}{2\,\sqrt{3}\,c+6\,s}\right\}$$

In[21]:=
```
Ntan[s_, c_] = Simplify[H1[[1]] + Npt[s, c] - H1[[5]]]
```

Out[21]=
$$\left\{\frac{-3+2\,\sqrt{3}\,c}{2\,\sqrt{3}\,c-6\,s},\ \frac{\sqrt{3}\,(1-2\,s)}{2\,\sqrt{3}\,c-6\,s}\right\}$$

Find the intersection of two circles of radius 1 with specified centres *a* and *b*. We single out the point of intersection that lies above the line from *a* to *b* if *a* is on the left.

In[22]:=
```
intC = icirc[{ax_, ay_}, {bx_, by_}] =
  Module[{diff, midpoint, magdiffsq},
   diff = {bx, by} - {ax, ay};
   midpoint = ({ax, ay} + {bx, by}) / 2;
   magdiffsq = diff.diff;
   midpoint + Sqrt[1/magdiffsq - 1/4] {-diff[[2]], diff[[1]]}
  ]
```

Out[22]=
$$\left\{\frac{ax+bx}{2}+(ay-by)\sqrt{-\frac{1}{4}+\frac{1}{(-ax+bx)^2+(-ay+by)^2}},\ \frac{ay+by}{2}+(-ax+bx)\sqrt{-\frac{1}{4}+\frac{1}{(-ax+bx)^2+(-ay+by)^2}}\right\}$$

Verify solution

In[23]:=
```
Simplify[(intC - {ax, ay}).(intC - {ax, ay})]
```

Out[23]= 1

In[24]:=
```
Simplify[(intC - {bx, by}).(intC - {bx, by})]
```

Out[24]= 1

Verify with a simple case that we have the *correct* solution, above the line from *a* to *b*

In[25]:=
```
icirc[{-1/2, 0}, {1/2, 0}]
```

Out[25]=
$$\left\{0,\ \frac{\sqrt{3}}{2}\right\}$$

Point *M* is the midpoint of the fourth side of the unrotated hexagon.

In[26]:=
```
Mpt = (H1[[4]] + H1[[5]]) / 2
```

Out[26]= $\left\{-\frac{1}{4}, -\frac{\sqrt{3}}{4}\right\}$

Normal to the mirror-line of symmetry that passes through point *M*

In[27]:=
```
Mnorm = Normalize[H1[[5]] - H1[[4]]]
```

Out[27]= $\left\{-\frac{\sqrt{3}}{2}, \frac{1}{2}\right\}$

Reflection in the mirror-line of symmetry that passes through point *M*

In[28]:=
```
refM = IdentityMatrix[2] - 2 Outer[Times, Mnorm, Mnorm];
```

Point *Q* is the reflection of the 10th intersection point between the hexagons in the mirror-line.

In[29]:=
```
Qpt[s_, c_] = Simplify[refM.pih[s, c][[10]], csAssume]
```

Out[29]= $\left\{\frac{3 - \sqrt{3}\ c - 3\ s}{2\sqrt{3}\ c - 6\ s}, \frac{-3\ c + \sqrt{3}\ (1 + s)}{2\sqrt{3}\ c - 6\ s}\right\}$

Point *W* is the intersection of circles of radius 1 centred at *M* and *O*

In[30]:=
```
Wpt0[s_, c_] = cst[Simplify[icirc[Mpt, Opt[s, c]]], 3] /.
  {Sqrt[x_] :> Sqrt[Together[x]]}
```

Out[30]=

$$\left\{\frac{2\sqrt{3}}{8c+8\sqrt{3}s} - \frac{3c}{8c+8\sqrt{3}s} + \frac{\sqrt{3}s}{8c+8\sqrt{3}s} - \frac{\sqrt{3}\sqrt{\frac{-37-2\sqrt{3}c+26c^2+10s-34\sqrt{3}cs}{-11+2\sqrt{3}c+6c^2-10s+2\sqrt{3}cs}}}{4\sqrt{3}c+12s} + \frac{3c\sqrt{\frac{-37-2\sqrt{3}c+26c^2+10s-34\sqrt{3}cs}{-11+2\sqrt{3}c+6c^2-10s+2\sqrt{3}cs}}}{2\left(4\sqrt{3}c+12s\right)} - \frac{\sqrt{3}s\sqrt{\frac{-37-2\sqrt{3}c+26c^2+10s-34\sqrt{3}cs}{-11+2\sqrt{3}c+6c^2-10s+2\sqrt{3}cs}}}{2\left(4\sqrt{3}c+12s\right)},\right.$$

$$\left.\frac{2\sqrt{3}}{8\sqrt{3}c+24s} - \frac{9c}{8\sqrt{3}c+24s} - \frac{5\sqrt{3}s}{8\sqrt{3}c+24s} + \frac{\sqrt{3}\sqrt{\frac{-37-2\sqrt{3}c+26c^2+10s-34\sqrt{3}cs}{-11+2\sqrt{3}c+6c^2-10s+2\sqrt{3}cs}}}{4c+4\sqrt{3}s} - \frac{c\sqrt{\frac{-37-2\sqrt{3}c+26c^2+10s-34\sqrt{3}cs}{-11+2\sqrt{3}c+6c^2-10s+2\sqrt{3}cs}}}{2\left(4c+4\sqrt{3}s\right)} + \frac{3\sqrt{3}s\sqrt{\frac{-37-2\sqrt{3}c+26c^2+10s-34\sqrt{3}cs}{-11+2\sqrt{3}c+6c^2-10s+2\sqrt{3}cs}}}{2\left(4c+4\sqrt{3}s\right)}\right\}$$

In[31]:=

$$\alpha 0 = \sqrt{\left(\frac{-37-2\sqrt{3}\,c+26\,c^2+10\,s-34\sqrt{3}\,c\,s}{-11+2\sqrt{3}\,c+6\,c^2-10\,s+2\sqrt{3}\,c\,s}\right)};$$

In[32]:=
```
α1 = Sqrt[Map[Collect[#, c] &,
    FullSimplify[Together[cst[α0^2, 2]]]]]
```

Out[32]=
$$\sqrt{\frac{-37+26\,c^2+10\,s-2\sqrt{3}\,c\,(1+17\,s)}{-11+6\,c^2-10\,s+2\sqrt{3}\,c\,(1+s)}}$$

In[33]:=
```
Wpt1[s_, c_] = Collect[Wpt0[s, c] /. {α0 → α}, α, Simplify]
```

Out[33]=
$$\left\{\frac{-3\,c+\sqrt{3}\,(2+s)}{8\left(c+\sqrt{3}\,s\right)}+\frac{\left(3\,c-\sqrt{3}\,(2+s)\right)\alpha}{8\left(\sqrt{3}\,c+3\,s\right)},\ \frac{-9\,c+\sqrt{3}\,(2-5\,s)}{8\left(\sqrt{3}\,c+3\,s\right)}+\frac{\left(-c+\sqrt{3}\,(2+3\,s)\right)\alpha}{8\left(c+\sqrt{3}\,s\right)}\right\}$$

In[34]:=
```
Wpt[s_, c_] = Wpt1[s, c] /. α → α1
```

Out[34]=
$$\left\{\frac{-3\,c+\sqrt{3}\,(2+s)}{8\left(c+\sqrt{3}\,s\right)}+\frac{\left(3\,c-\sqrt{3}\,(2+s)\right)\sqrt{\frac{-37+26\,c^2+10\,s-2\sqrt{3}\,c\,(1+17\,s)}{-11+6\,c^2-10\,s+2\sqrt{3}\,c\,(1+s)}}}{8\left(\sqrt{3}\,c+3\,s\right)},\ \frac{-9\,c+\sqrt{3}\,(2-5\,s)}{8\left(\sqrt{3}\,c+3\,s\right)}+\frac{\left(-c+\sqrt{3}\,(2+3\,s)\right)\sqrt{\frac{-37+26\,c^2+10\,s-2\sqrt{3}\,c\,(1+17\,s)}{-11+6\,c^2-10\,s+2\sqrt{3}\,c\,(1+s)}}}{8\left(c+\sqrt{3}\,s\right)}\right\}$$

Point *Y* is the intersection of circles of radius 1 centred at *N* and *Q*

In[35]:=
```
Ypt0[s_, c_] =
  cst[Simplify[icirc[Npt[s, c], Qpt[s, c]]], 3] /.
   {Sqrt[x_] :> Sqrt[Together[x]]}
```

Out[35]=
$$\left\{0,\ \frac{\sqrt{3}}{2\sqrt{3}\,c-6\,s}-\frac{3\,c}{2\sqrt{3}\,c-6\,s}+\frac{\sqrt{3}\,s}{2\sqrt{3}\,c-6\,s}+\frac{3\sqrt{\frac{3+\sqrt{3}\,c-3\,c^2+3\,s-5\sqrt{3}\,c\,s}{3-\sqrt{3}\,c-c^2-3\,s+\sqrt{3}\,c\,s}}}{2\sqrt{3}\,c-6\,s}-\frac{\sqrt{3}\,c\sqrt{\frac{3+\sqrt{3}\,c-3\,c^2+3\,s-5\sqrt{3}\,c\,s}{3-\sqrt{3}\,c-c^2-3\,s+\sqrt{3}\,c\,s}}}{2\sqrt{3}\,c-6\,s}-\frac{3\,s\sqrt{\frac{3+\sqrt{3}\,c-3\,c^2+3\,s-5\sqrt{3}\,c\,s}{3-\sqrt{3}\,c-c^2-3\,s+\sqrt{3}\,c\,s}}}{2\sqrt{3}\,c-6\,s}\right\}$$

In[36]:=
$$\beta 0=\sqrt{\frac{3+\sqrt{3}\,c-3\,c^2+3\,s-5\sqrt{3}\,c\,s}{3-\sqrt{3}\,c-c^2-3\,s+\sqrt{3}\,c\,s}};$$

In[37]:=
```
β1 = Sqrt[Map[Collect[#, c] &,
    FullSimplify[Together[cst[β0^2, 2]]]]]
```

Out[37]=
$$\sqrt{\frac{-3\,c^2+\sqrt{3}\,c\,(1-5\,s)+3\,(1+s)}{3-c^2+\sqrt{3}\,c\,(-1+s)-3\,s}}$$

In[38]:=
```
Ypt1[s_, c_] = Collect[Ypt0[s, c] /. {β0 → β}, β, Simplify]
```

Out[38]=
$$\left\{0,\ \frac{-3\,c+\sqrt{3}\,(1+s)}{2\sqrt{3}\,c-6\,s}+\frac{\left(3-\sqrt{3}\,c-3\,s\right)\beta}{2\sqrt{3}\,c-6\,s}\right\}$$

In[39]:=
```
Ypt[s_, c_] = Ypt1[s, c] /. β → β1
```

Out[39]=
$$\left\{0,\ \frac{\left(3-\sqrt{3}\,c-3\,s\right)\sqrt{\frac{-3\,c^2+\sqrt{3}\,c\,(1-5\,s)+3\,(1+s)}{3-c^2+\sqrt{3}\,c\,(-1+s)-3\,s}}}{2\sqrt{3}\,c-6\,s}+\frac{-3\,c+\sqrt{3}\,(1+s)}{2\sqrt{3}\,c-6\,s}\right\}$$

Point X is the intersection of circles of radius 1 centred at *N* and *O*

In[40]:=
```
Xpt0[s_, c_] =
  cst[Simplify[icirc[Npt[s, c], Opt[s, c]]], 3] /.
   {Sqrt[x_] :> Sqrt[Together[x]]}
```

Out[40]=
$$\left\{-\frac{3\,s}{-6+8\,c^2}+\frac{2\sqrt{3}\,c\,s}{-6+8\,c^2}+\frac{\sqrt{3}\sqrt{\frac{12+5\sqrt{3}\,c-44\,c^2-4\sqrt{3}\,c^3+32\,c^4}{6-5\sqrt{3}\,c-4\,c^2+4\sqrt{3}\,c^3}}\,s}{-6+8\,c^2}-\frac{2\,c\sqrt{\frac{12+5\sqrt{3}\,c-44\,c^2-4\sqrt{3}\,c^3+32\,c^4}{6-5\sqrt{3}\,c-4\,c^2+4\sqrt{3}\,c^3}}\,s}{-6+8\,c^2},\ \frac{\sqrt{3}}{-6+8\,c^2}+\frac{c}{-6+8\,c^2}-\frac{2\sqrt{3}\,c^2}{-6+8\,c^2}-\frac{3\sqrt{3}\sqrt{\frac{12+5\sqrt{3}\,c-44\,c^2-4\sqrt{3}\,c^3+32\,c^4}{6-5\sqrt{3}\,c-4\,c^2+4\sqrt{3}\,c^3}}}{2\left(-3\sqrt{3}+4\sqrt{3}\,c^2\right)}+\frac{3\,c\sqrt{\frac{12+5\sqrt{3}\,c-44\,c^2-4\sqrt{3}\,c^3+32\,c^4}{6-5\sqrt{3}\,c-4\,c^2+4\sqrt{3}\,c^3}}}{2\left(-3\sqrt{3}+4\sqrt{3}\,c^2\right)}+\frac{\sqrt{3}\,c^2\sqrt{\frac{12+5\sqrt{3}\,c-44\,c^2-4\sqrt{3}\,c^3+32\,c^4}{6-5\sqrt{3}\,c-4\,c^2+4\sqrt{3}\,c^3}}}{-3\sqrt{3}+4\sqrt{3}\,c^2}\right\}$$

In[41]:=
$$\gamma 0=\sqrt{\frac{12+5\sqrt{3}\,c-44\,c^2-4\sqrt{3}\,c^3+32\,c^4}{6-5\sqrt{3}\,c-4\,c^2+4\sqrt{3}\,c^3}};$$

In[42]:=
```
γ1 = Sqrt[Map[Collect[#, c] &,
    FullSimplify[Together[cst[γ0^2, 2]]]]]
```

Out[42]=
$$\sqrt{\frac{4+7\sqrt{3}\,c+8\,c^2}{2+\sqrt{3}\,c}}$$

In[43]:=
```
Xpt1[s_, c_] = Collect[Xpt0[s, c] /. {γ0 → γ},
   γ, Simplify[cst[Simplify[#], 3]] &]
```

Out[43]=
$$\left\{\frac{\left(-3+2\sqrt{3}\,c\right)s}{-6+8\,c^2}+\frac{\left(\sqrt{3}-2\,c\right)s\,\gamma}{-6+8\,c^2},\ \frac{\sqrt{3}+c-2\sqrt{3}\,c^2}{-6+8\,c^2}+\frac{\left(-3+\sqrt{3}\,c+2\,c^2\right)\gamma}{-6+8\,c^2}\right\}$$

In[44]:= `Xpt[s_, c_] = Xpt1[s, c] /. γ → γ1`

Out[44]= $\left\{\frac{\left(-3+2\sqrt{3}\,c\right)s}{-6+8c^2}+\frac{\left(\sqrt{3}-2c\right)\sqrt{\frac{4+7\sqrt{3}\,c+8c^2}{2+\sqrt{3}\,c}}\;s}{-6+8c^2},\ \frac{\left(-3+\sqrt{3}\,c+2c^2\right)\sqrt{\frac{4+7\sqrt{3}\,c+8c^2}{2+\sqrt{3}\,c}}}{-6+8c^2}+\frac{\sqrt{3}+c-2\sqrt{3}\,c^2}{-6+8c^2}\right\}$

Dot product between the vectors YW and YQ to determine whether the angle ∠WYQ is greater than 90 degrees, i.e. this needs to be **negative** for a valid cover.

In[45]:=
```
dp1[s_, c_] = Collect[Collect[FullSimplify[
    cst[FullSimplify[Together[(Wpt1[s, c] - Ypt1[s, c]).
        (Qpt[s, c] - Ypt1[s, c])]], 3]], {α, β},
   FullSimplify] /. {β^2 → β1^2}, {α, β}, FullSimplify]
```

Out[45]= $-\left(\left(\left(c\left(c\left(3+4c\left(c+\sqrt{3}(-2+s)\right)\right)-9\sqrt{3}(-1+s)\right)+9(-1+s)\right)\left(-3c^2+\sqrt{3}\,c\,(1-5s)+3(1+s)\right)\right)\Big/\left(2\left(3-4c^2\right)^2\left(3-c^2+\sqrt{3}\,c\,(-1+s)-3s\right)\right)\right)+\frac{3+6c^2-3s+\sqrt{3}\,c\,(-5+2s)}{-48+64c^2}+\frac{1}{16\left(3-4c^2\right)^2}\left(-21\sqrt{3}(-1+s)+c\left(-81+66s+2c\left(\sqrt{3}(3-2s)-2c\left(-19+6\sqrt{3}\,c+6s\right)\right)\right)\right)\beta+\alpha\left(\frac{c\left(5-2\sqrt{3}\,c-2s\right)+\sqrt{3}(-1+s)}{-48+64c^2}+\frac{\left(3-10c^2-3s+\sqrt{3}\,c\,(3+2s)\right)\beta}{-48+64c^2}\right)$

In[46]:= `dp[s_, c_] = dp1[s, c] /. {α → α1, β → β1};`

Dot product between the vectors WM and WY to determine whether the angle the angle ∠MWY is greater than 90 degrees, i.e. this needs to be **negative** for a valid cover.

In[47]:=
```
dp2[s_, c_] =
  Together[(Mpt - Wpt[s, c]).(Ypt[s, c] - Wpt[s, c])];
```

Vertices of the reflected triangle *B*′. The original triangle *B* has vertices B_1, B_2, B_3 in clockwise order; these correspond to the second vertex of the unrotated hexagon, and intersection points 3 and 2. So the vertices of the reflected triangle in **counterclockwise** order are:

In[48]:=
```
BpVerts[s_, c_] = Simplify[
  Map[refM.# &, {H1[[2]], pih[s, c][[3]], pih[s, c][[2]]}]]
```

Out[48]=
$$\left\{\left\{0, \frac{1}{\sqrt{3}}\right\}, \left\{\frac{3-2\sqrt{3}\ c}{2\sqrt{3}\ c+6\ s}, \frac{\sqrt{3}\ (1+2\ s)}{2\sqrt{3}\ c+6\ s}\right\}, \left\{\frac{-3+2\sqrt{3}\ c}{2\sqrt{3}\ c-6\ s}, \frac{\sqrt{3}\ (1-2\ s)}{2\sqrt{3}\ c-6\ s}\right\}\right\}$$

In[49]:=
```
drawBp[σ_] := Module[{s, c, bpv, marg},
  s = Sin[σ];
  c = Cos[σ];
  bpv = BpVerts[s, c];
  marg = 0.02;
  Graphics[
   {
    Table[
     {Line[{bpv[[i]], bpv[[Mod[i + 1, 3, 1]]]}],
      Text["B′"i, bpv[[i]],
       {{-0.5, -1.2}, {1.5, 0}, {-1.5, 0}}[[i]]]}
     , {i, 1, 3}],
    {PointSize[.0075],
     {Point[Xpt[s, c]],
      Text["X", Xpt[s, c], {1, -1.2}]},
     {Point[Wpt[s, c]], Text["W", Wpt[s, c], {0, -1.2}]},
     {Point[Ypt[s, c]], Text["Y", Ypt[s, c], {-1, 1.2}]},
     Line[{Wpt[s, c], Ypt[s, c]}],
     Circle[Opt[s, c], 1],
     Circle[Npt[s, c], 1]
    }
   }
   , BaseStyle → 18, ImageSize → {600, 300},
   PlotRange -> {{bpv[[2, 1]] - marg, bpv[[3, 1]] + marg},
     {Min[bpv[[2, 2]], bpv[[3, 2]] - marg],
      bpv[[1, 2]] + marg}}
  ]]
```

In[50]:=
```
drawBp[1.3 Degree]
```

Out[50]=



Outwards-pointing normals for the sides of triangle *B*′

In[51]:=
```
BpNorms[s_, c_] = Simplify[cst[Simplify[
    Table[RotationMatrix[π / 2].(BpVerts[s, c][[i]] -
        BpVerts[s, c][[Mod[i + 1, 3, 1]]]), {i, 1, 3}]], 3]]
```

Out[51]= $\left\{\left\{\frac{3-2\sqrt{3}\,c}{6\,c+6\sqrt{3}\,s},\ \frac{-3+2\sqrt{3}\,c}{2\sqrt{3}\,c+6\,s}\right\},\ \left\{\frac{\left(\sqrt{3}-2\,c\right)s}{-3+4\,c^2},\ \frac{\left(\sqrt{3}-2\,c\right)c}{-3+4\,c^2}\right\},\ \left\{\frac{-3+2\sqrt{3}\,c}{6\left(c-\sqrt{3}\,s\right)},\ \frac{-3+2\sqrt{3}\,c}{2\sqrt{3}\,c-6\,s}\right\}\right\}$

Values on sides of dot products with these normals

In[52]:=
```
BpDP[s_, c_] =
  Simplify[cst[Simplify[Table[BpNorms[s, c][[i]].
      BpVerts[s, c][[i]], {i, 1, 3}]], 3]]
```

Out[52]= $\left\{\frac{-3+2\sqrt{3}\,c}{6\left(c+\sqrt{3}\,s\right)},\ \frac{\sqrt{3}-2\,c}{-6+8\,c^2},\ \frac{-3+2\sqrt{3}\,c}{6\left(c-\sqrt{3}\,s\right)}\right\}$

Check whether a point lies inside the triangle *B*′

In[53]:=
```
insideBp[s_, c_, x_] := Apply[And,
   Table[BpNorms[s, c][[i]].x < BpDP[s, c][[i]], {i, 1, 3}]]
```

Polygon for the third reduction: this joins up all the vertices of the cover, but needs two chord areas added to it, between O_{tan} and *W* and between N_{tan} and *Y*.

```
In[54]:= red3[s_, c_] = Module[{rh, ph},
    rh = rotH[s, c];
    ph = pih[s, c];
    {Ypt[s, c], Ntan[s, c],
     H1[[2]], ph[[4]], ph[[5]], H1[[4]], ph[[8]],
     ph[[9]], H1[[6]], Otan[s, c], Wpt[s, c]}
   ];
```

Find the two chord lengths

```
In[55]:= chord1[s_, c_] =
   Sqrt[(Wpt[s, c] - Otan[s, c]).(Wpt[s, c] - Otan[s, c])];

In[56]:= chord2[s_, c_] =
   Sqrt[(Ypt[s, c] - Ntan[s, c]).(Ypt[s, c] - Ntan[s, c])];
```

Total area of cover is area of polygon plus areas between chords and circles

```
In[57]:= coverArea[s_, c_] = areaPoly[red3[s, c]] +
    chordArea[chord1[s, c]] + chordArea[chord2[s, c]];
```

Nominal rotation angle from paper: 1.3 degrees

```
In[58]:= σNom = (13 / 10) Degree
```

Out[58]= $\frac{13\,°}{10}$

```
In[59]:= N[coverArea[Sin[σNom], Cos[σNom]], 100]

Out[59]= 0.844115376859376746806104420762830688615710154306004199⋱
  2611141990036756345193458767424887453255339710
```

Check that dot products are negative

```
In[60]:= N[{dp[s, c], dp2[s, c]} /.
   {s → Sin[σNom], c → Cos[σNom]}, 100]

Out[60]= {-5.079985563427177518068920278297475162948895692489732 1⋱
      63623765912546571144980969292302144510337483652 × 10^-6,
  -0.0266619221388340312360319010776318864187493132494797⋱
    81375749873327327086871230884672828204494596303 70}
```

Check that points W, X and Y are inside the triangle B'

```
In[61]:= Map[insideBp[Sin[σNom], Cos[σNom], #] &,
  {Wpt[Sin[σNom], Cos[σNom]],
   Xpt[Sin[σNom], Cos[σNom]], Ypt[Sin[σNom], Cos[σNom]]}]

Out[61]= {True, True, True}
```

Plots around this value

In[62]:=
```
Plot[coverArea[Sin[σ * Degree], Cos[σ * Degree]],
  {σ, 1.2, 1.4}]
```

Out[62]=



In[63]:=
```
Plot[dp[Sin[σ * Degree], Cos[σ * Degree]], {σ, 1.2, 1.4}]
```

Out[63]=



In[64]:=
```
Plot[dp2[Sin[σ * Degree], Cos[σ * Degree]], {σ, 1.2, 1.4}]
```

Out[64]=



Find a value closer to the minimum possible

In[65]:=
```
N[Sin[σNom]]
```

Out[65]= 0.0226873

In[66]:=
```
s0 = N[s /. FindRoot[Evaluate[dp[s, √(1 - s²)]],
      {s, 22 / 1000}, WorkingPrecision → 2000], 20]
```

Out[66]= 0.022589436005758246174

In[67]:=
```
ArcSin[s0] / Degree
```
Out[67]= 1.2943894447036010115

We round up to get a safe value

In[68]:=
$$\sigma 0 = \frac{1\,294\,389\,444\,703\,601\,012}{10^{18}}\ \text{Degree};$$

In[69]:=
```
N[coverArea[Sin[σ0], Cos[σ0]], 100]
```
Out[69]= 0.84411529712841905921419219237658670360968648515129381296740128625425787576791426299593654853617214 47

Check that dot products are negative

In[70]:=
```
N[{dp[s, c], dp2[s, c]} /. {s → Sin[σ0], c → Cos[σ0]}, 100]
```
Out[70]= {-4.3518604836559154004332576721487197864037217898329499963262606074575553002453278580318104378490027 56 × 10^-22,
-0.02666630545360558533349865825325377308750142412268585551116387468684189101234536032048712272117600922}

Check that points *W*, *X* and *Y* are inside the triangle *B′*

In[71]:=
```
Map[insideBp[Sin[σ0], Cos[σ0], #] &,
  {Wpt[Sin[σ0], Cos[σ0]],
   Xpt[Sin[σ0], Cos[σ0]], Ypt[Sin[σ0], Cos[σ0]]}]
```
Out[71]= {True, True, True}

See if we can find an explicit formula for the critical angle where the first dot product is zero.

In[72]:=
```
dpsc0 = Map[Collect[#, {α, β}, Simplify] &,
   Factor[cst[Collect[Numerator[
        Together[Collect[Numerator[Together[dp1[s, c]]],
           {α, β}, Simplify] /. {β^2 → β1^2}]],
       {α, β}], 3], Extension → Automatic]]
```
Out[72]=
$$-\left(\sqrt{3}-2c\right)^2 \Big(-108c^4+18(-1+s)-147\sqrt{3}\,c(-1+s)+4\sqrt{3}\,c^3(-33+17s)-3c^2(-51+40s)+\Big(12\sqrt{3}\,c^4-42\sqrt{3}(-1+s)+21c(-1+s)-4c^3(-5+3s)+\sqrt{3}\,c^2(-53+32s)\Big)\beta+\alpha\Big((-3+4c^2)\left(c+\sqrt{3}\,c^2+2\sqrt{3}(-1+s)-cs\right)+\Big(9c^2+4c^4+18(-1+s)+21\sqrt{3}\,c(-1+s)-4\sqrt{3}\,c^3(-5+3s)\Big)\beta\Big)\Big)$$

In[73]:=
```
dpsc1 = Collect[Collect[Collect[Numerator[
      Together[α^2 - α1^2 /. Solve[dpsc0 == 0, α][[1]]]],
     β, cstNumerator[#, 3] &], β] /.
   {β^2 → β1^2}, β, cstNumerator[#, 3] &]
```

Out[73]=
$$\left(16\left(\sqrt{3}-2c\right)^2\left(1188+3708\sqrt{3}\,c+12915c^2+4179\sqrt{3}\,c^3-29859c^4-23972\sqrt{3}\,c^5+16532c^6+21392\sqrt{3}\,c^7+304c^8-5312\sqrt{3}\,c^9-1088c^{10}-1188s-3708\sqrt{3}\,c\,s-13509c^2s-6033\sqrt{3}\,c^3s+22956c^4s+20492\sqrt{3}\,c^5s-6896c^6s-12144\sqrt{3}\,c^7s-1472c^8s+1344\sqrt{3}\,c^9s\right)\right)\Big/\left(3-c^2+\sqrt{3}\,c\,(-1+s)-3s\right)-32\left(\sqrt{3}-2c\right)^2\left(180\sqrt{3}+1710c+1041\sqrt{3}\,c^2-1455c^3-2008\sqrt{3}\,c^4-1220c^5+720\sqrt{3}\,c^6+960c^7+64\sqrt{3}\,c^8-180\sqrt{3}\,s-1710c\,s-1131\sqrt{3}\,c^2s+600c^3s+1420\sqrt{3}\,c^4s+1280c^5s-192\sqrt{3}\,c^6s-320c^7s\right)\beta$$

In[74]:=
```
dpsc2 =
  Collect[cst0[Collect[cst[Numerator[Together[β^2 - β1^2 /.
        Solve[dpsc1 == 0, β][[1]]]], 3], c]], s, Simplify]
```

Out[74]=
$$2822688-10373184\sqrt{3}\,c-353018736c^2-711908352\sqrt{3}\,c^3-412816014c^4+2601277524\sqrt{3}\,c^5+6110860779c^6-2690055324\sqrt{3}\,c^7-13659887394c^8-891700128\sqrt{3}\,c^9+13735058592c^{10}+3808190528\sqrt{3}\,c^{11}-6801052768c^{12}-2957901824\sqrt{3}\,c^{13}+1370876672c^{14}+967281664\sqrt{3}\,c^{15}+45142528c^{16}-115122176\sqrt{3}\,c^{17}-39256064c^{18}+311296\sqrt{3}\,c^{19}+1269760c^{20}+2\left(-1411344+5186592\sqrt{3}\,c+175803696c^2+358547472\sqrt{3}\,c^3+294133437c^4-1120716702\sqrt{3}\,c^5-2886476418c^6+829811907\sqrt{3}\,c^7+5434404768c^8+743278248\sqrt{3}\,c^9-4485924384c^{10}-1484626864\sqrt{3}\,c^{11}+1672726784c^{12}+847541504\sqrt{3}\,c^{13}-171283968c^{14}-189776640\sqrt{3}\,c^{15}-37322752c^{16}+10381312\sqrt{3}\,c^{17}+5349376c^{18}+372736\sqrt{3}\,c^{19}\right)s$$

In[75]:=
```
dpsc3 = Factor[Numerator[
    Together[s^2 + c^2 - 1 /. Solve[dpsc2 == 0, s][[1]]]],
   Extension → Automatic]
```

Out[75]=
$$\left(\sqrt{3} - 2\,c\right)^8 c^{12} \left(\sqrt{3} + 2\,c\right)^4 \left(-15\,549\,831 - 6\,053\,016\,\sqrt{3}\,c + 2\,315\,081\,568\,c^2 + 12\,649\,847\,712\,\sqrt{3}\,c^3 + 93\,993\,026\,304\,c^4 + 127\,164\,667\,776\,\sqrt{3}\,c^5 + 262\,892\,593\,920\,c^6 + 13\,489\,688\,064\,\sqrt{3}\,c^7 - 409\,701\,622\,272\,c^8 - 343\,084\,558\,336\,\sqrt{3}\,c^9 - 298\,136\,715\,264\,c^{10} + 78\,429\,970\,432\,\sqrt{3}\,c^{11} + 288\,979\,156\,992\,c^{12} + 105\,668\,182\,016\,\sqrt{3}\,c^{13} + 58\,874\,068\,992\,c^{14} + 5\,688\,655\,872\,\sqrt{3}\,c^{15} + 800\,653\,312\,c^{16}\right)$$

The best we can do is express the cosine of the critical angle as a root of a degree-16 polynomial. With a change of variable, we can give the polynomial integer coordinates.

In[76]:=
```
dpsc4 =
  Factor[Numerator[Together[dpsc3[[-1]] /. {c → χ / √3}]]]
```

Out[76]=
$$-102\,022\,441\,191 - 39\,713\,837\,976\,\chi + 5\,063\,083\,389\,216\,\chi^2 + 27\,665\,216\,946\,144\,\chi^3 + 68\,520\,916\,175\,616\,\chi^4 + 92\,703\,042\,808\,704\,\chi^5 + 63\,882\,900\,322\,560\,\chi^6 + 3\,277\,994\,199\,552\,\chi^7 - 33\,185\,831\,404\,032\,\chi^8 - 27\,789\,849\,225\,216\,\chi^9 - 8\,049\,691\,312\,128\,\chi^{10} + 2\,117\,609\,201\,664\,\chi^{11} + 2\,600\,812\,412\,928\,\chi^{12} + 951\,013\,638\,144\,\chi^{13} + 176\,622\,206\,976\,\chi^{14} + 17\,065\,967\,616\,\chi^{15} + 800\,653\,312\,\chi^{16}$$

In[77]:=
```
ArcCos[χ / √3] / Degree /.
  NSolve[dpsc4 == 0, Reals, WorkingPrecision → 100]
```

Out[77]=
```
{109.4057290066882256170273350746585510978146464714886671857372188341769546727729778159479960313546920 7,
 86.4031475760421832749891702562986700383071755977603233975448719344956252538484656019048513468863220 06,
 2.08191576010178649209267193342811365856022666221134738948570430156095632027074877160620164402295 2,
 1.29438944470360101151937639612367734552213704024507726520550965918290588547146342076989782045825 4}
```